

IDL Tutorial



AULA 08

COMPREENSÃO DE CÓDIGO TRANSPOSIÇÃO LOOP X VETOR

Transposição 01



**MODELO DE DECAIMENTO DE POTENCIA DO
SINAL DE TRANSMISSÃO SEM FIO.**

MODELO DE OKUMURA HATA

```
; Programa simulador de potencia de sinal  
; Utiliza modelo de Okumura Hata
```

```
PRO con01
```

```
X = dindgen(146) ;Distância da antena em km  
Y = X  
POT = 16 ;Potencia de transmissão em dB  
f = 800 ;Frequencia em MHz  
hb = 100 ; Altura da antena transmissora em metros  
hm = 10 ; Altura da antena receptora em metros
```

```
FOR i=0, 145 DO BEGIN  
    X[i] = X[i]/10 + 1.5  
ENDFOR
```

```
FOR i=0, 145 DO BEGIN  
    Y[i] = POT - ( 69.55 + 26.16*alog10(f) - 13.82*alog10(hb) - $  
        ( 0.8 + (1.1*alog10(f) - 0.7)*hm - 1.56*alog10(f) ) + $  
        (44.9 - 6.55*alog10(hb))*alog10(X[i]) )  
ENDFOR
```

```
PLOT, X, Y
```

```
END
```

```
; Programa simulador de potencia de sinal  
; Utiliza modelo de Okumura Hata
```

```
PRO con01b
```

```
X = dindgen(146)/10+1.5 ;Distância da antena em km
```

```
POT = 16 ;Potencia de transmissão em dB
```

```
f = 800 ;Frequencia em MHz
```

```
hb = 100 ; Altura da antena transmissora em metros
```

```
hm = 10 ; Altura da antena receptora em metros
```

```
Y = POT - ( 69.55 + 26.16*alog10(f) - 13.82*alog10(hb) - $  
    ( 0.8 + (1.1*alog10(f) - 0.7)*hm - 1.56*alog10(f) ) + $  
    (44.9 - 6.55*alog10(hb))*alog10(X) )
```

```
PLOT, X, Y
```

```
END
```

Transposição 02



**AMOSTRAGEM DE SINAL POR PARTE
POSITIVA E PARTE NEGATIVA EM GRÁFICOS
SEPARADOS.**

PRO con02

```
X = dindgen(100)
Y1 = X
Y2 = X

FOR i=0, 99 DO BEGIN
    temp = 10*randomn(SEED)*cos(X[i])
    IF temp GT 0 THEN BEGIN
        Y1[i] = temp
        Y2[i] = 0
    ENDIF ELSE BEGIN
        Y1[i] = 0
        Y2[i] = -temp
    ENDELSE
ENDFOR

PLOT, X, Y1
WINDOW, 1
PLOT, X, Y2

END
```

PRO con02b

X = dindgen(100)

RAND = randomn(SEED,100)

Y = 10*RAND*cos(X)

PLOT, X, Y*(Y **GT** 0)

window, 1

PLOT, X, -Y*(Y **LT** 0)

END

Transposição 03



**CALCULO DE DISTÂNCIA ENTRE UM PONTO
INICIAL E UM CONJUNTO DE PONTOS.**

PRO con03

```
pa = [1.2,2.5]
```

```
pi = randomn(SEED, 2, 50)
```

```
d = dindgen(50)
```

```
FOR i=0, 49 DO BEGIN
```

```
    d[i] = sqrt( (pa[0]-pi[0,i])^2 + $  
                (pa[1]-pi[1,i])^2 )
```

```
ENDFOR
```

```
PLOT, dindgen(50), d
```

```
END
```

PRO con03b

```
pa = [1.2,2.5]
pi = randomn(SEED, 2, 50)
d = dindgen(50)
aux2 = [1,1]

a1 = pa#transpose(d^0)
a2 = (a1 - pi)^2
d = sqrt( a2##transpose(aux2) )

PLOT, dindgen(50), d
```

END

Transposição 04



**GRÁFICO DE POSIÇÃO DE PROJÉTIL
LANÇADO DA ORIGEM.**

PRO con04

```
vox = 5  
voy = 10  
a = -9.8
```

```
Sx = dindgen(101)  
Sy = Sx
```

```
FOR i=0, 100 DO BEGIN  
    Sx[i] = Sx[i]/100  
    Sx[i] = Sx[i] * (-2*Vox*voy)/a  
    Sy[i] = (a/(2*Vox^2))*Sx[i]^2 + (Voy/Vox)*Sx[i]  
ENDFOR
```

```
PLOT, Sx, Sy*(Sy GT 0)
```

END

PRO con04b

vox = 5

voy = 10

a = -9.8

Sx = (dindgen(101)/100) * (-2*Vox*Voy)/a

Sy = (a/(2*Vox^2)) * Sx^2 + (Voy/Vox) * Sx

PLOT, Sx, Sy*(Sy **GT** 0)

END

Transposição 05



**EXIBIÇÃO DE GRÁFICOS DE MÓDULO DE
ÂNGULO DE UM CONJUNTO DE VETORES.**

PRO con05

```
M = dindgen(100)
P = M
FOR i=0, 99 DO BEGIN
    x = randomn(SEED)
    y = randomn(SEED)
    M[i] = sqrt(x^2+y^2)
    P[i] = atan(y,x)
END

PLOT, dindgen(100), M
WINDOW, 1
PLOT, dindgen(100), P

END
```

PRO con05b

X = randomn (SEED, 100)

Y = randomn (SEED, 100)

M = sqrt (X^2+Y^2)

P = atan (Y, X)

PLOT, dindgen (100) , M

WINDOW, 1

PLOT, dindgen (100) , P

END

Transposição 06



**GRÁFICO DE SUPERFÍCIE DE UMA ONDA SE
DISSIPANDO A PARTIR DE UM PONTO
CENTRAL.**

PRO con06

X = dindgen(101)-50

Y = X

Z = dindgen(101,101)

FOR i=0, 100 **DO BEGIN**

FOR j=0, 100 **DO BEGIN**

 d = sqrt(X[j]^2+Y[i]^2)

 Z[j,i] = cos(d)/(d+2)

ENDFOR

ENDFOR

SURFACE, Z

END

PRO con06b

$X = \text{dindgen}(101) - 50$

$X = X \# X^0$

$Y = \text{transpose}(X)$

$d = \text{sqrt}(X^2 + Y^2)$

$Z = \cos(d) / (d + 2)$

SURFACE, Z

END

Dúvidas?



[HTTP://IDLTUTORIAL.BLOGSPOT.COM](http://idltutorial.blogspot.com)

ANTONIOPAULOVP@GMAIL.COM

LUCIOMARASSI@GMAIL.COM